
本日のスケジュール

• Arduino（Mozzi）で音を出すプログラムを作ろう



質問コーナー

• 何も知らない状態からJava言語やC言語のようなプログラミング言語を勉強す
るのは用語集のようなものを扱って覚えて行くものなのでしょうか?


• 言語によらず使いまわせる知識はある。（if文のような基本的な概念とか、
アルゴリズムとか。）ここは一度学べばあまり陳腐化しない


• 言語依存の知識は頻繁にアップデートされるので、調べて追いつく力の方
が重要


• ググったりChatGPTに聞くための語彙を増やすためには根本的な知識がいる



とりあえずArduinoで音を出す





ピエゾスピーカーなら 
ここの1k抵抗は無し 
ダイナミックスピーカー 
なら適宜抵抗値を調整 

（5Wとか超えたらアンプを
使うようにする）



https://akizukidenshi.com/catalog/g/gK-08217/　

https://akizukidenshi.com/catalog/g/gK-08217/




const int button_pin = 2; 
const int sound_pin = 9; 

void setup() { 
  pinMode(button_pin, INPUT); 
  //tone()を使うときは3,11以外 
  pinMode(sound_pin, OUTPUT); 
} 

void loop() { 
  if (digitalRead(2) == HIGH) { 
    tone(sound_pin, 1000); 
  } else { 
    noTone(sound_pin); 
  } 
  delay(20); 
}

tone_minimal.ino



音名と周波数

• 周波数のままでは音楽用に使いづらい


• 中心のラ（A4）＝440Hz　が一般的なチューニング


• 1オクターブ＝周波数が2倍(A5=880Hz、A6=1760Hz)


• 半音上がるのを12回繰り返すと周波数が2倍になる(＝半音で ずつ上昇)


• A4を69番として、音名に番号を振る（MIDIノート番号）

21/12

Frequency(Hz) = 440 ⋅ 2(midi−69/12)



const int button_pin = 2; 
const int sound_pin = 9; 

bool is_playing = false; 
int button_prev= LOW; 
float freq = 440; 

float midiToFreq(int midi){ 
  return 440.*pow(2.,(midi-69.)/12.); 
} 

void setup() { 
  pinMode(button_pin,INPUT); 
  pinMode(sound_pin, OUTPUT); 
} 

void loop() { 
  auto button_state = digitalRead(button_pin); 

  if(button_prev == LOW && button_state==HIGH){ 
    is_playing = !is_playing; 
    freq = midiToFreq(random(40,100)); 
  } 
  if(is_playing){ 
    tone(sound_pin,freq); 
  }else{ 
    noTone(sound_pin); 
  } 

  delay(20); 
  button_prev = button_state; 
}

tone_random_flip.ino

Frequency(Hz) = 440 ⋅ 2(midi−69/12)



Mozziで高度な音作り



Pulse Code Modulation(PCM)

Aquegg~commonswiki, CC-BY-SA 3.0 , https://commons.wikimedia.org/wiki/File:Pcm.svg 

1サンプルごとのビット 
（この例では4ビット）を 
アナログ電圧に直す変換器

(DAコンバーター)が必要

https://commons.wikimedia.org/wiki/File:Pcm.svg


Pulse Density Modulation
縦軸が粗い代わりに横軸を細かくする

Moxfyre, CC 表示-継承 3.0, https://commons.wikimedia.org/w/index.php?curid=3239064 による

ノイズは含まれているが、
高周波に偏っている。 
コンデンサーで高周波を 
取り除くだけで 
そのまま使える。 

（ノイズ成分が20kHz以上
で聞こえなければ 
それすら不要）

https://commons.wikimedia.org/w/index.php?curid=3239064


Mozzi

ArduinoのタイマーやPWM機能を最大限に活用して、DACなしで複雑な音声合成をするライブラリ





Mozziの注意点

• スピーカーへ出力するピンは9番で固定


• 他のPWMピン（analogWrite）も基本的に使えなくなる


• analogRead()の代わりにmozziAnalogRead()を使う必要がある


• なので、内部的にanalogRead()を呼ぶADCTouchのようなライブラリとの組
み合わせも難しい


• delay()の代わりにEventDelayクラスを使う必要がある



Mozziの使い方

• setup()とloop()に加えて、void updateControl()とint updateAudio()という関数
を作る


• 遅い処理（センサーの読み取りなど）はupdateControl()に


• updateAudioでreturnする値が1サンプルごとの電圧（音圧）になる


• この値の範囲は  -244～243でなければならない


• loopの中ではaudioHook()という関数だけを呼ぶ



#include <MozziGuts.h> 
#include <Oscil.h> 
#include <tables/sin2048_int8.h> 

Oscil <SIN2048_NUM_CELLS, AUDIO_RATE> aSin(SIN2048_DATA); 

void setup(){ 
  startMozzi(); 
  aSin.setFreq(440); 
} 

void updateControl(){ 

} 

int updateAudio(){ 
  return aSin.next(); 
} 

void loop(){ 
  audioHook(); 
}

mozzi_sine.ino



エンベロープを理解しよう

鍵盤押す 鍵盤離す



エンベロープを理解しよう

Attack Decay

Sustain

Release

Mozziの場合整数で0~243



#include <MozziGuts.h> 
#include <mozzi_midi.h> 
#include <Oscil.h> 

#include <ADSR.h> 
#include <tables/sin8192_int8.h> 

Oscil<8192, AUDIO_RATE> aOscil(SIN8192_DATA); 

//エンベロープをかけるためのクラス 
ADSR<AUDIO_RATE, AUDIO_RATE> envelope;         

unsigned int Dur, Atk, Dec, Sus, Rel;  
int button_prev = LOW; 

void setup() { 
  startMozzi(1024); 
  Atk = 10; 
  Dec = 10; 
  Sus = 50; 
  Rel = 100; 
  envelope.setTimes(Atk, Dec, Sus, Rel); 
  envelope.setADLevels(255, 128); 
} 

void updateControl() { 
  auto button_state = digitalRead(2); 
  if (button_prev == LOW && button_state == HIGH) { 
    auto f = mtof((int)random(65, 100)); 
    aOscil.setFreq(f); 
    envelope.noteOn(); 
  } 
  if (button_prev == HIGH && button_state == LOW) { 
    envelope.noteOff(); 
  } 
  button_prev = button_state; 
} 

int updateAudio() { 
  envelope.update(); 
  return envelope.next() * aOscil.next() / 255; 
} 

void loop() { 
  audioHook(); 
}

mozzi_envelope.ino



http://sheep-me.me/2019/10/24/geidai_16/ 
ひつじさんのサンプルより、Steve Reich「Piano Phase」の再現

http://sheep-me.me/2019/10/24/geidai_16/


#include <MozziGuts.h> 
#include <mozzi_midi.h> 
#include <Oscil.h> 
#include <EventDelay.h> 
#include <ADSR.h> 
#include <tables/sin8192_int8.h> 

Oscil <8192, AUDIO_RATE> aOscil(SIN8192_DATA); 
Oscil <8192, AUDIO_RATE> aOscil2(SIN8192_DATA); 
ADSR <AUDIO_RATE, AUDIO_RATE> envelope_A; 
ADSR <AUDIO_RATE, AUDIO_RATE> envelope_B; 
unsigned int Dur, Atk, Dec, Sus, Rel; 

//piano phaseのパターン 
unsigned int pattern[] = {64, 66, 71, 73, 74, 66, 
64, 73, 71, 66, 74, 73}; 

//二つのタイマーを用意 
int phase_A = 0; 
int phase_B = 0; 
EventDelay timer_B; 
EventDelay timer_A; 

void setup() { 
  startMozzi(1024); 
  Atk = 10; 
  Dec = 10; 
  Sus = 50; 
  Rel = 100; 
  envelope_A.setTimes(Atk, Dec, Sus, Rel); 
  envelope_A.setADLevels(255, 128); 
  envelope_B.setTimes(Atk, Dec, Sus, Rel); 
  envelope_B.setADLevels(255, 128); 

  timer_A.set(150); 
  timer_B.set(151); 
  timer_B.start(); 
  timer_A.start(); 
} 

 

void updateControl() 
{   
  if (timer_A.ready()) 
  { 
    int note = pattern[phase_A]; 
    phase_A = (phase_A + 1) % 12; 
    aOscil.setFreq(mtof(note)); 
    envelope_A.noteOn(); 
    timer_A.start(); 
  } 

  if (timer_B.ready()) 
  { 
    int note = pattern[phase_B]; 
    phase_B = (phase_B + 1) % 12; 
    aOscil2.setFreq(mtof(note)); 
    envelope_B.noteOn(); 
    timer_B.start(); 
  } 
} 

int updateAudio() 
{ 
  envelope_A.update(); 
  envelope_B.update(); 
  int A = (envelope_A.next() * aOscil.next()) >> 8; 
  int B = (envelope_B.next() * aOscil2.next()) >> 8; 

  return (A + B) / 2; 
} 

void loop() { 
  audioHook(); 
} 

mozzi_reich.ino



より高度なことがやりたい人は



https://www.electro-smith.com/daisy 

https://www.electro-smith.com/daisy


https://bela.io/

https://bela.io/


参考資料



http://sheep-me.me/講義資料/　
電子工作創作表現 / 音楽音響創造特殊研究 講義資料(2019),ひつじ

http://sheep-me.me/%e8%ac%9b%e7%be%a9%e8%b3%87%e6%96%99/


MozziWS, 中西宣人
https://yoshihito-nakanishi.com/workshops/mozzi 

https://yoshihito-nakanishi.com/workshops/mozzi


https://yoppa.org/ssaw14/ 
SSAW14 – サウンド&ソフトウェアアート 2014, 田所淳

https://yoppa.org/ssaw14/


https://speakerdeck.com/tomoyanonymous/a-bit-high-layer-audio-embedded-platforms-in-2020 

https://speakerdeck.com/tomoyanonymous/a-bit-high-layer-audio-embedded-platforms-in-2020


残りの授業のスケジュール

• 6/16 課題制作展示相談会1


• 6/23 デジタルファブリケーション①(銅箔コイルでスピーカー制作)


• 6/30 課題制作展示相談会2


• 7/07 デジタルファブリケーション②(レーザーカッターで可変抵抗制作)


• 7/14 課題作品インストール作業


• 7/21 講評&振り返り&撤収 ＋、7月に電子基板発注WSを助手の川田さんと
一緒にやるかもしれません



課題（再掲）
あなたにとっての「パーソナルコンピューター」を作れるようになること

• 例えば：自動ドローイングマシン、アクセサリー、入力装置


• Or、授業の中で学んだ内容のうちのトピックのいずれかを教えるための道具
作り


• 例えば、Conditional Designの新しいルールを考えるとかもあり◎


• 自由枠：自分の課題制作等に授業内で学んだ内容を反映する 

• 技術的な難易度よりも、なぜそれを作ったのか伝わるように



課題のためのルール

• 最終課題の製作に必要な材料等は原則自身で用意すること


• Arduinoは授業終了時まで借りていっても良い。（無くさないように）


• ラボ内で半田ごて等の作業道具が必要な場合は松浦かラボスタッフに声をか
けること。



Making Things Move ―動くモノを作るためのメカニズムと材料の基本 (Make: PROJECTS) (2012), 
Dustyn Roberts (著), 岩崎 修 (監修), 金井 哲夫 (翻訳)



Prototyping Lab 第2版 ―「作りながら考える」ためのArduino実践レシピ (Make: PROJECTS),小林 茂 (2017)



Arduinoをはじめよう 第4版 (Make: PROJECTS) (2023),Massimo Banzi (著), Michael Shiloh (著), 船田 巧 (翻訳)


